

第八章 CMOS模拟集成电路

8.6 CMOS共源放大电路


CMOS共源放大电路

- ◆ 共源放大电路是CMOS集成电路中最基础的放大极形式
- ◆ 无源电阻负载CMOS共源放大电路
 - 电路结构
 - 直流大信号电压转移特性
 - 电路分析方法
 - 增益计算与仿真
- ◆ 有源负载CMOS共源放大电路
 - MOS二极管负载
 - MOS电流源负载

无源电阻负载CMOS共源放大电路

NMOS共源放大电路

$$v_o = V_{DD} - i_D R_D$$



直流转移特性

◆ 输入电压：0~5V，步长0.1V

◆ $V_{DD}=5V$

◆ M1: 10um/1um, $R_D=2k\Omega$

```
.title CS_AMP_DC
```

```
M1 2 1 0 0 n08 W=10U L=1U
```

```
RD 3 2 2k
```

```
VDD 3 0 DC=5
```

```
Vin 1 0 DC=2
```

```
.OP
```

```
.DC Vin 0 5 0.1
```


```
.probe v(2) v(1) i(M1)
```

```
.option post probe
```

```
*.MODEL 语句省略
```


```
.end
```

*直流工作点仿真
*直流扫描仿真




直流转移特性

M1截止区




确定直流工作点



交流放大概念

Q点处, $| \text{切线斜率} | > 1$
交流信号得到了放大




输入信号范围

输入 $< 0.7V$,
M1截止区,
输出最大5V,
饱和

输入交流分量
幅度 $< 0.5V$

输入 $> 2.5V$,
M1线性区,
输出最小0.5V,
饱和



M1应该工作在饱和区,
线性区无法实现放大



M1应该避免进入线性区,
否则最小输出电压饱和

电路分析方法

1, 选定直流工作点, 为电路提供合适偏置, 保证MOS管工作在饱和区

◆ 通过直流扫描仿真


- 根据直流转移特性, 确定输入直流偏置电压取2V



电路分析方法

2, 确定MOS管交流小信号模型,
得到对应的小信号模型参数

◆ 通过OP仿真



低频MOS管交流小信号模型
(不考虑背栅效应)




$g_m = 1.65\text{mS}$
 $g_{ds} = 38.4\mu\text{S}$
 $r_{ds} = 26\text{k}\Omega$

```
subckt
element 0:m1
model 0:n08
region Saturati
id 1.0701m
ibs 0.
ibd -28.5986f
vgs 2.0000
vds 2.8599
vbs 0.
vth 700.0000m
vdsat 1.3000
vod 1.3000
beta 1.2664m
gam eff 400.0000m
gm 1.6463m
gds 38.4091u
gmb 393.5324u
cdtot 2.2910f
cgtot 21.1559f
cstot 18.1174f
cbtot 747.3930a
cgs 18.1174f
cgd 2.2910f
```

电路分析方法

3, 得到电路的等效交流小信号模型

- ◆ 将MOS管用交流小信号模型替代
- ◆ 所有对应直流电压的节点改为接地



电路分析方法

4, 基于等效交流小信号模型, 列写电路方程求解, 得到期望的电路特性


◆ 关于节点D列写KCL方程

$$\frac{v_o}{R_D} + \frac{v_o}{r_{ds}} + g_m v_{gs} = 0$$

$$v_i = v_{gs}$$

◆ 增益

$$A_v = \frac{v_o}{v_i} = -g_m \frac{1}{\frac{1}{r_{ds}} + \frac{1}{R_D}}$$



增益公式直观解释

$$i_1 = -g_m v_{gs} = -g_m v_i$$


$$r_D = r_{ds} \parallel R_D$$

$$v_o = i_1 r_D$$

◆ 增益

$$A_v = \frac{v_o}{v_i} = -g_m (r_{ds} \parallel R_D)$$

- MOS管跨导
- 节点D输出电阻



增益估算

$$A_v = -g_m (r_{ds} \parallel R_D)$$

◆ 快速估算

$$r_{ds} \parallel R_D \approx R_D$$

$$A_v \approx -g_m R_D = -1.65 \times 2 = -3.3$$

◆ 常规估算

$$A_v = -1.65(26 \parallel 2) = -3.06$$

■ ~9.7dB

基于OP仿真结果

gm=1.65mS
gds=38.4uS
rds=26kΩ

增益仿真 (AC仿真)

- ◆ CL: 后级电路的负载效应

```
.title CS_AMP_AC
```

```
M1 2 1 0 0 n08 W=10U L=1U
```

```
RD 3 2 2k
```

```
VDD 3 0 DC=5
```

```
Vin 1 0 DC=2 AC=1
```

```
CL 2 0 1p
```

```
.OP
```

```
.AC DEC 10 10 1g
```


```
.probe vdb(2) vp(2)
```

```
.option post probe
```

```
*.MODEL 语句省略
```


```
.end
```

*直流工作点仿真
*交流仿真



增益仿真 (AC仿真)

交流仿真是
线性仿真



为观察非线性，必须改用瞬态仿真，且输入交流信号幅度足够大

TRAN仿真

```
.title CS_AMP_TRAN
```

```
M1 2 1 0 0 n08 W=10U L=1U
```

```
RD 3 2 2k
```

```
VDD 3 0 DC=5
```

```
Vin 1 0 sin(2 0.6 1k 0 0 0)
```

```
CL 2 0 1p
```

```
.OP
```

```
.TRAN 1u 2m
```

```
.probe v(1) v(2)
```

```
.option post probe
```


```
*.MODEL 语句省略
```

```
.end
```

*直流工作点仿真
*瞬态仿真


◆ 输入信号：幅度0.6V

- M1会进入线性区
- 最小输出饱和




TRAN仿真

输入直流分量2V,
交流分量0.6V



输出直流分量2.86V



饱和, 波形明显失真

输入大信号, 非线性
导致增益变小

如果提高增益

- ◆ 增益公式


$$A_v = -g_m (r_{ds} \parallel R_D)$$

- ◆ 负载电阻增大为25kΩ

$$\begin{aligned} A_v &= -1.65(26 \parallel 25) \\ &= -21.03 \end{aligned}$$


- 增益增大为21

- ◆ 直流工作点必须重新选择




有源负载CMOS共源放大电路

无源电阻负载CMOS共源放大电路



$$i_1 = -g_m v_{gs} = -g_m v_i$$

$$r_D = r_{ds} \parallel R_D$$




$$A_v = \frac{v_o}{v_i} = -g_m (r_{ds} \parallel R_D)$$

提高增益? $g_m \nearrow$ $r_D \nearrow$

MOS管有源负载


◆ 无源电阻负载 R_D -> MOS管有源负载

◆ 二极管接法的MOS管



PMOS二极管负载



◆ 电流源接法的MOS管



PMOS电流源负载

PMOS二极管负载


◆ PMOS二极管等效电路



◆ 等效电阻

$$i_x = g_m v_{gs} + \frac{v_x}{r_{ds}}$$

$$v_{gs} = v_x$$



$$R_D = \frac{v_x}{i_x} = \frac{1}{g_{m2}} \parallel r_{ds2}$$

PMOS二极管负载

$$R_D = \frac{1}{g_{m2}} \parallel r_{ds2}$$


◆ 增益

$$\begin{aligned} A_v &= -g_{m1} (r_{ds1} \parallel R_D) \\ &= -g_{m1} \left(r_{ds1} \parallel r_{ds2} \parallel \frac{1}{g_{m2}} \right) \end{aligned}$$


◆ 一般情况下, $\frac{1}{g_m} \ll r_{ds}$

$$g_{m1} = 1.65\text{mS}, \quad \frac{1}{g_{m1}} = 0.6\text{k}\Omega, \quad r_{ds1} = 26\text{k}\Omega$$

$$r_{ds1} \parallel r_{ds2} \parallel \frac{1}{g_{m2}} \approx \frac{1}{g_{m2}}$$



$$A_v \approx -\frac{g_{m1}}{g_{m2}} \quad \text{跨导的比值}$$



PMOS二极管负载


- ◆ 长沟道近似下、简单直流MOS管模型

$$g_m = \left. \frac{\partial I_D}{\partial V_{GS}} \right|_Q = \sqrt{2\mu_0 C_{ox} \frac{W}{L} I_D}$$

- M1/M2电流相等

$$A_v \approx -\frac{g_{m1}}{g_{m2}}$$


$$= -\frac{\sqrt{2\mu_n C_{ox} \left(\frac{W}{L}\right)_1 I_D}}{\sqrt{2\mu_p C_{ox} \left(\frac{W}{L}\right)_2 I_D}} = -\sqrt{\frac{\mu_n \left(\frac{W}{L}\right)_1}{\mu_p \left(\frac{W}{L}\right)_2}}$$



设置合适的宽长比
可以实现所需的增益

PMOS电流源负载

◆ PMOS电流源等效电路




◆ 等效电阻

$$v_{gs} = 0$$

$$g_m v_{gs} = 0$$

$$R_D = \frac{v_x}{i_x} = r_{ds2}$$

◆ 增益



$$\begin{aligned} A_v &= -g_{m1} (r_{ds1} \parallel R_D) \\ &= -g_{m1} (r_{ds1} \parallel r_{ds2}) \end{aligned}$$


具有更高的增益

两种有源负载对比

	PMOS二极管有源负载	PMOS电流源有源负载
◆ 增益	$A_v = -g_{m1} \left(r_{ds1} \parallel r_{ds2} \parallel \frac{1}{g_{m2}} \right)$	$A_v = -g_{m1} (r_{ds1} \parallel r_{ds2})$
◆ 输入电阻	$R_i = \infty$	$R_i = \infty$
◆ 输出电阻	$R_o = r_{ds1} \parallel r_{ds2} \parallel \frac{1}{g_{m2}} \approx \frac{1}{g_{m2}}$	$R_o = r_{ds1} \parallel r_{ds2}$

PMOS电流源负载共源放大器仿真

- ◆ $V_{DD}=5V$
- ◆ $M_1, M_2: 10\mu m/1\mu m$
- ◆ $V_b=2.4V$



OP仿真与DC仿真

```
.title CS_AMP_DC
* with current source load
M1 2 1 0 0 n08 W=10U L=1U
M2 2 4 3 3 p08 W=10U L=1U
Vbias 4 0 DC=2.4
VDD 3 0 DC=5
Vin 1 0 DC=2
```


```
.OP
.DC Vin 0 5 0.1
.probe v(2) v(1) i(M1)
```

```
.option post probe
```

```
*.MODEL 语句省略
```

```
.end
```


*直流工作点仿真
*直流扫描仿真



- ◆ 根据OP仿真结果，M1工作于饱和区

直流转移特性

- ◆ 直流工作点选择
 - 输入电压: 2V
 - 输出电压: ~2.4V



增益估算

$$\begin{aligned}
 A_v &= -g_{m1} (r_{ds1} \parallel r_{ds2}) \\
 &= -g_{m1} \left(\frac{1}{g_{ds1} + g_{ds2}} \right) \\
 &= -1.62 \left(\frac{10^3}{38.4 + 46.4} \right) \\
 &= -19
 \end{aligned}$$

◆ 25.6dB

subckt		
element	0:m1	0:m2
model	0:n08	0:p08
region	Saturati	Saturati
id	1.0510m	-1.0510m
ibs	0.	0.
ibd	-23.6220f	26.3780f
vgs	2.0000	-2.6000
vds	2.3622	-2.6378
vbs	0.	0.
vth	700.0000m	-700.0000m
vdsat	1.3000	-1.9000
vod	1.3000	-1.9000
beta	1.2437m	582.2478u
gam eff	400.0000m	570.0000m
gm	1.6169m	1.1063m
gds	38.4091u	46.4249u
gmb	386.5029u	352.5018u
cdtot	2.2752f	2.2843f
cgtot	21.1400f	21.2083f
cstot	18.1174f	18.1832f
cbtot	747.3930a	740.7470a
cgs	18.1174f	18.1832f
cgd	2.2752f	2.2843f

AC仿真

```
.title CS_AMP_AC
* with current source load
M1 2 1 0 0 n08 W=10U L=1U
M2 2 4 3 3 p08 W=10U L=1U
Vbias 4 0 DC=2.4
VDD 3 0 DC=5
Vin 1 0 DC=2 AC=1
```

```
CL 2 0 1p
```

```
.OP
.AC DEC 10 10 1g
.probe vdb(2) vp(2)
```


```
.option post probe
```

```
*.MODEL 语句省略
```

```
.end
```

*直流工作点仿真
*交流仿真

◆ CL: 后级电路的负载效应



AC仿真


◆ 反相放大

◆ 低频增益

■ 25.6dB

◆ 单位增益带宽

■ 251MHz



TRAN仿真

```
.title CS_AMP_TRAN  
* with current source load  
M1 2 1 0 0 n08 W=10U L=1U  
M2 2 4 3 3 p08 W=10U L=1U  
Vbias 4 0 DC=2.4  
VDD 3 0 DC=5  
Vin 1 0 sin(2 1m 1k 0 0 0)
```

```
CL 2 0 1p
```

```
.OP  
.TRAN 1u 2m  
.probe v(1) v(2)
```


```
.option post probe
```

```
*.MODEL 语句省略
```

```
.end
```

*直流工作点仿真
*瞬态仿真


- ◆ 输入正弦信号: $DC=2V$
 $AC=1mV$ $Freq=1kHz$




TRAN仿真

- ◆ 反相放大
- ◆ 输入信号幅度
 - 1mV
- ◆ 输出信号幅度
 - 19mV
- ◆ 增益
 - 19倍
 - 25.6dB

输入信号波形



输出信号波形



$$2.343 - 2.362 = 0.019 V$$

小结

- ◆ CMOS共源放大电路及其仿真
 - 基于HSPICE
 - 对电路特性有全面的了解
- ◆ 采用共源共栅结构
 - 输出电阻可以进一步提高
 - 实现更大的电压增益